@echo off
:: enable /V:ON flag

set _j_test_delay_variable=Y
set _j_test_delay_variable=!_j_test_delay_variable!
if NOT %_j_test_delay_variable%==Y (
	cmd /V:ON /C %~f0 %*
	exit /B
)
set _j_test_delay_variable=
setlocal enableextensions
setlocal enabledelayedexpansion

set j_script_dir=%~dp0
set j_jenny_root_dir=!j_script_dir!
set j_current_dir=!cd!
pushd !j_jenny_root_dir!

call common\function.bat  init

set j_command=!j_jenny_root_dir!/!j_os_name!/%1.bat

if exist !j_command! (
	call common\function.bat pack_args j_command_args 1 %*
	call !j_command! !j_command_args!
) else (
	call common\function.bat pack_args j_command_args 0 %*
	call !j_jenny_root_dir!/!j_os_name!/help.bat !j_command_args!
)

popd
exit /B !ERRORLEVEL!
